

A KIM-1 SIDEREAL/SOLAR CLOCK

by John O. Bumgarner

This program includes both major and minor timing loops and can be tuned to run accurately with a wide range of clock frequencies. This program will also fit nicely into the KIM-1 scratch pad area.

Adjusting the clock rate by time fillers is an example of "fixing it in the software" rather than electronically adjusting the crystal clock rate. The fillers at locations 0216 and 0218 provide a 4 μ sec. to 12 μ sec. range of delays depending on the instructions loaded here. A 1 μ sec. change in this delay results in about 19.82 seconds per day change in the clock rate. Inspection of the *MCS 650X Instruction Set Summary Card* shows several 2-byte instructions with execution times of from 2 μ sec. to 6 μ sec. The instructions chosen should not modify the time stored at locations 0020, 0021, 0022, the X register or the decimal mode bit.

The minor timing loop has the delay formula

$$\text{Delay} = 6 + 5(N-1) + V \mu\text{sec. where}$$

$$6 = 2 + 4, 2 \mu\text{sec. for the load X,}$$

$$4 \mu\text{sec. for the loop without a branch}$$

$$5 = \text{loop time with a branch, } \mu\text{sec.}$$

$$N = \text{loop count, } \$01 \text{ to } \$FF \text{ or } \$00$$

$$V = \text{variable time filler } 2 \mu\text{sec. to } 6 \mu\text{sec.}$$

The delay introduced by this timing loop will vary from 8 μ sec. ($N = 1, V = 2 \mu$ sec.) to 1282 μ sec. ($N = 255, V = 6 \mu$ sec.). The loop will go 1287 μ sec. if $N = 00$ since X is decremented to 255 before testing.

An example: Suppose you determine that your KIM-1 is running 10.63 seconds/day fast by observing it against a time standard such as WWV. Patient observation over several weeks will be necessary if the KIM-1 is close to the desired rate. The number of microseconds to delay per second is:

$$\frac{10.63 \times 10^6}{60 \times 60 \times 24} = \frac{10630000}{86400} = 123.03 \text{ or } 123 \mu\text{sec. per second}$$

The loop parameters are found this way:


$$123 - 6 = 117 \text{ (the fixed delay amount)}$$

$$\frac{117}{5} = 23.4 \text{ (23 loops plus } 0.4 \times 5 = 2 \mu\text{sec. more),}$$

$$23_{10} = 17_{16}$$

The loop parameters would be \$17 at location 021E and the filler would be 2 μ sec. — say a LDY immediate instruction. By choosing the loop count and filler appropriately any delay in the range of the loop can be achieved to the nearest microsecond. A change of one microsecond in this loop will result in a change in the clock rate of 0.0864 seconds per day or about 30 seconds per year.

The KIM-1 programmable internal timers could be used for both the major and minor delays but require "watching" the timer or the wiring of the interrupt lines on the KIM-1 and are not so simple to tune as the delay loops.



The crystal clock rate can be accurately adjusted to the nearest microsecond by "fixing" it in the KIM-1 software fillers.

CLOCK PROGRAM FOR KIM-1

HEX ADDR	CODE	LABEL		COMMENTS
0200	A2E9	L1	LDX \$E9	Load X with timing loop count, Note 1.
2	d8		CLD	Clear decimal mode.
3	A520	L2	LDA HR	} Move time to display area. Note 2
5	85Fb		STA Fb	
7	A521		LDA MI	
9	85FA		STA FA	
b	A522		LDA SE	
d	85F9		STA F9	
020F	8623		STX T1	Save loop count. Note 2
0211	20 1F 1F		JSR SCANDS	Jump to KIM-1 display subroutine.
4	A623		LDX T1	restore loop count.
6	A100		LDA (0,X)	} Note 3.
8	A100		LDA (0,X)	
A	CA		DEX	decrement loop count.
b	d0E6		BNE L2	Branch to loop head if X ≠ 0.
d	A216		LDX \$90	Load X with timing loop count. Note 6.
021F	CA		DEX	Decrement loop count.
0220	d0Fd		BNE --1	If X ≠ 0, go back one line.
2	b500		LDA 0,X	>Note 4.
4	F8		SED	Set decimal mode for arithmetic.
5	18		CLC	Clear carry bit.
6	A522		LDA SE	Pick up seconds count.
8	6901		ADC \$01	Add one to it.
A	8522		STA SE	Put it back.
C	C960		CMP \$60	Compare seconds count to sixty.
023E	d0d0		BNE L1	If seconds not equal to sixty, to the start. Note 5.
0	8622		STX SE	Set seconds count to zero (X is zero here).
2	A521		LDA MI	Pick up minutes count.
4	6900		ADC \$00	Add one to it (the CMP at 022C set the carry bit on equal).
6	8521		STA MI	Put it back.
0238	C960		CMP \$60	Compare seconds count to sixty.
A	d0C4		BNE L1	If minutes not equal to sixty, go to start.
C	8621		STX MI	Set minutes to zero.
023E	A520		LDA HR	Pick up the hours count.
0240	6900		ADC \$00	Add one to it.
2	8520		STA HR	Put it back.
4	C924		CMP \$24	Compare hours to 24. Note 6.
6	d0b8		BNE L1	If hours not equal to 24, go to the start.
8	8620		STX HR	Set hours to zero.
024A	F064		BEQ L1	Go to start.
			END @0200	

major timing loop

minor timing loop

establish & clean environment

NOTES:

- For regular time rate (UT or solar rate) load \$EA instead of \$E9.
- Zero page locations used: 0020 for seconds, 0021 for minutes, 0022 for hours and 0023 temporary.
- These two instructions are timing fillers. The two above are a 12μsec delay for sidereal time. For solar time my KIM-1 needs a delay of 8μsec and I use E6 23 A523 (INC T1, LDA T1) here.
- This instruction is a time filler. It will probably be different in your KIM-1 for sidereal or solar rate, see text.
- The compare instruction just before this branch sets the carry bit on equal compare. Therefore when seconds equal sixty the branch is not taken and adding zero later results in adding one due to the carry.
- For solar time on my KIM-1 this constant is \$16 instead of \$90.
- For 12 hour clock use the following code:

0244	C913		CMP \$13	Compare hours to thirteen.
6	dob8		BNE L1	If hours not equal to thirteen, go to the start.
8	E8		INX	increment X. X is not \$01.
9	8620		STX HR	Store X in hours.
024b	Fob3		BNE L1	Go to start.
			END @0200	

- To run, load the time, HR:MI:SE, in 20 to 22 on page zero and GO at address 0200.